
;-----------------------------------------------------------------
;Przykładowe procedury obsługi wyświetlacza ze sterownikiem HD44780
;-----------------------------------------------------------------
;-----------------------------------------------------------------
;Konfiguracja portow - wyswietlacz podlaczony 4 liniami danych
;-----------------------------------------------------------------
  DB4   EQU   P2.4
  DB5   EQU   P2.5
  DB6   EQU   P2.6
  DB7   EQU   P2.7
  EN    EQU   P1.0
  RW    EQU   P1.1
  RS    EQU   P1.2
  Dane  EQU   P2
;-----------------------------------------------------------------
;Odczyt dwoch czworek
;-----------------------------------------------------------------
LCD_odczyt_4:
  setb  RW          ;odczyt
  orl   Dane,#0F0h  ;linie do LCD na 1
  setb  EN          ;impuls na linii E
  mov   A,Dane      ;odczyt bardziej znaczacej czworki
  clr   EN
  anl   A,#0F0h     ;zamaskowanie mlodszej czesci ACC
  mov   R7,A        ;zapamietanie starszej czworki
  setb  EN
  mov   A,Dane      ;odczyt mlodszej czworki
  clr   EN
  anl   A,#0F0h     ;zamaskowanie starszej czesci ACC
  swap  A           ;mlodsza czworka na swoje miejsce
  orl   A,R7        ;scalenie czworek
  ret
;-----------------------------------------------------------------
;Zapis dwoch czworek
;-----------------------------------------------------------------
LCD_zapis_4:
  push  ACC         ;zachowanie ACC
  clr   RW          ;zapis
  orl   Dane,#0F0h  ;starsze bity portu na "1" (4..7 <- 1)
  orl   A,#0Fh      ;bity 0..3 pozostaja bez zmian
  anl   Dane,A      ;wystawienie na port starszej czworki
  setb  EN
  clr   EN
  pop   ACC         ;odtworzenie ACC
  swap  A           ;zamiana czworek
  orl   Dane,#0F0h  ;starsze bity portu na "1" (4...7 <- 1)
  orl   A,#0Fh      ;bity 0..3 pozostaja bez zmian
  anl   Dane,A      ;wystawienie na port mlodszej czworki
  setb  EN
  clr   EN
  ret
;-----------------------------------------------------------------
;Oczekiwanie na gotowosc do przyjecia nastepnego polecenia
;-----------------------------------------------------------------
LCD_czekaj:
  clr   RS             ;rejestr sterujacy
  setb  RW             ;odczyt
  acall LCD_odczyt_4   ;odczytanie dwoch czworek
  jb    ACC.7,LCD_czekaj ;jesli bit7 = 1 zajety
  ret



;-----------------------------------------------------------------
;Inicjalizacja
;-----------------------------------------------------------------
LCD_init:
  clr   RS         ;rejestr sterujacy
  clr   RW         ;zapis
  clr   EN
  setb  EN
  mov   Dane,#28h  ;0 0 1 0 0 0 0 0
               ;| | | | | | ---- 00 – bity niewykorzystane
               ;| | | | | ------ 0 – (F) Font, 0: znaki 5x8

             ;| | | | -------- 0 – (N) Number of lines, 0: 1 linia
             ;| | | ---------- 0 - (DL) Data Length, 0: 4 bitowy interfejs
             ;---------------- 001 – polecenie Function Set

  clr   EN
  acall LCD_czekaj   ;odczekanie na gotowosc
  mov   A,#28h
  acall LCD_zapis_4
  acall LCD_czekaj
  mov   A,#0Eh     ;0 0 0 0 1 1 1 0
               ;| | | | | | | -- 0 – (B) Blinking, 0: wylaczone

             ;| | | | | | ---- 1 – (C) Cursor On, 1: wlaczony
             ;| | | | | ------ 1 - (D) Display On, 1: wlaczony
             ;---------------- 00001 – polecenie Display OnOff Control

  acall LCD_zapis_4
  acall LCD_czekaj
  mov   A,#06h     ;0 0 0 0 0 1 1 0

             ;| | | | | | | -- 0 - (S) Shift, 0: brak przesuwania
               ;| | | | | | ---- 0 – (I/D) Inc/Dec, 1: inkrementacja

             ;---------------- 000001 – polecenie Entry mode Set
  acall LCD_zapis_4
  acall LCD_czekaj
  ret
;-----------------------------------------------------------------
;Wyczyszczenie
;-----------------------------------------------------------------
LCD_ClrScr:
  clr   RS
  mov   A,#01h     ;0 0 0 0 0 0 0 1
               ;| | | | | | | |

             ;---------------- 00000001 – polecenie ClrScr
  acall LCD_zapis_4
  acall LCD_czekaj
  ret
;-----------------------------------------------------------------
;Wypisanie znaku (znak podawany w akumulatorze)
;-----------------------------------------------------------------
LCD_znak:
  setb  RS
  acall LCD_zapis_4
  acall LCD_czekaj
  ret
;-----------------------------------------------------------------
;Koniec
;-----------------------------------------------------------------


